


AD-A167 560


MICROCOMPUTER OPTIMIZATION OF FEED EXCITATION VOLTAGES
TO OBTAIN DESIRED ANTENNA RADIATION PATTERNS(U) NAVAL
OCEAN SYSTEMS CENTER SAN DIEGO CA J W ROCKWAY ET AL.
FEB 86 NOSC/TD-866 F/G 9/5

1/1

UNCLASSIFIED

NL





MICROCOPY

CHART

2

TD 866

A167 560

Technical Document 866

February 1986

MICROCOMPUTER OPTIMIZATION OF FEED EXCITATION VOLTAGES TO OBTAIN DESIRED ANTENNA RADIATION PATTERNS

John W. Rockway

James C. Logan

Peder M. Hansen

DTIC
ELECTE
MAY 12 1986
S D



Naval Ocean Systems Center

San Diego, California 92152-5000

Approved for public release; distribution is unlimited.

DTIC FILE COPY

86 5 12 123

NAVAL OCEAN SYSTEMS CENTER

San Diego, California 92152-5000

F. M. PESTORIUS, CAPT, USN
Commander

R. M. HILLY,
Technical Director

ADMINISTRATIVE INFORMATION

This work was performed by members of the Antenna and RF Systems Integration Branch, NAVOCEANSYSCEN, for the Space and Naval Warfare Systems Command, Washington, DC, and the US Army Communications Electronics Command, Fort Monmouth, NJ.

Released by
I.C. Olson, Head
Antenna and RF Systems
Integration Branch

Under authority of
G.E. Ereckson, Head
Shipboard Systems Division

DISCLAIMER NOTICE

THIS DOCUMENT IS BEST QUALITY PRACTICABLE. THE COPY FURNISHED TO DTIC CONTAINED A SIGNIFICANT NUMBER OF PAGES WHICH DO NOT REPRODUCE LEGIBLY.

REPORT DOCUMENTATION PAGE

1a REPORT SECURITY CLASSIFICATION UNCLASSIFIED		1b RESTRICTIVE MARKINGS	
2a SECURITY CLASSIFICATION AUTHORITY		3 DISTRIBUTION / AVAILABILITY OF REPORT Approved for public release; distribution is unlimited.	
2b DECLASSIFICATION / DOWNGRADING SCHEDULE			
4 PERFORMING ORGANIZATION REPORT NUMBER(S) NOSC TD 866		5 MONITORING ORGANIZATION REPORT NUMBER(S)	
6a NAME OF PERFORMING ORGANIZATION Naval Ocean Systems Center	6b OFFICE SYMBOL <i>(if applicable)</i> Code 822	7a NAME OF MONITORING ORGANIZATION	
6c ADDRESS <i>(City, State and ZIP Code)</i> San Diego, CA 92152-5000		7b ADDRESS <i>(City, State and ZIP Code)</i>	
8a NAME OF FUNDING / SPONSORING ORGANIZATION Space and Naval Warfare Systems Command	8b OFFICE SYMBOL <i>(if applicable)</i> SPAWAR-614	9 PROCUREMENT INSTRUMENT IDENTIFICATION NUMBER	
8c ADDRESS <i>(City, State and ZIP Code)</i> Space and Naval Warfare Systems Command Washington, DC 20363-5100		10 SOURCE OF FUNDING NUMBERS	
		PROGRAM ELEMENT NO 62543N	PROJECT NO X1-43454100
		TASK NO 822-CM41	Agency Accession No. DN088 509
11 TITLE <i>(include Security Classification)</i> MICROCOMPUTER OPTIMIZATION OF FEED EXCITATION VOLTAGES TO OBTAIN DESIRED ANTENNA RADIATION PATTERNS			
12 PERSONAL AUTHOR(S) John W. Rockway, James C. Logan, Peder M. Hansen			
13a TYPE OF REPORT Final	13b TIME COVERED FROM _____ TO _____	14 DATE OF REPORT <i>(Year, Month, Day)</i> February 1986	15 PAGE COUNT 23
16 SUPPLEMENTARY NOTATION			
17 COSATI CODES		18 SUBJECT TERMS <i>(Continue on reverse if necessary and identify by block number)</i>	
FIELD	GROUP	SUB GROUP	antenna, array, MININEC, patterns, design, optimal feed excitation, microcomputer techniques, numerical techniques, method of moments, optimization, least squares
19 ABSTRACT <i>(Continue on reverse if necessary and identify by block number)</i> Recently a computer program, MININEC, has become available that can be used to calculate antenna characteristics, including the actual current distribution on the array elements. This program is generally available and can be run on various microcomputers. A design procedure for broadcast array antennas has been developed which uses this computer program to calculate antenna characteristics for a specified array geometry. These results are then used by a separate routine that optimizes the excitation voltages, by gradient search, in order to achieve, as nearly as possible, a specified pattern.			
20 DISTRIBUTION / AVAILABILITY OF ABSTRACT <input type="checkbox"/> UNCLASSIFIED UNLIMITED <input checked="" type="checkbox"/> SAME AS RPT <input type="checkbox"/> DTIC USERS		21 ABSTRACT SECURITY CLASSIFICATION UNCLASSIFIED	
22a NAME OF RESPONSIBLE INDIVIDUAL John W. Rockway, James C. Logan, Peder M. Hansen		22b TELEPHONE <i>(include Area Code)</i> (619) 225-2646	22c OFFICE SYMBOL Code 822

CONTENTS

Introduction . . . 1

MININEC . . . 2

Power Gain . . . 2

Optimization Algorithm . . . 4

Sample Problem . . . 5

Summary and Recommendations . . . 8

References . . . 9

APPENDIX: Key Variables and Program Listing of OPTIMAL . . . 10



Accession For	
NTIS CRA&I	<input checked="" type="checkbox"/>
DTIC TAB	<input type="checkbox"/>
Unannounced	<input type="checkbox"/>
Justification	
By _____	
Distribution / _____	
Availability Codes	
Dist	Avail and/or Special
A-1	23

INTRODUCTION

Properly designed antenna arrays can be used to control or shape radiation patterns for covert applications, to ensure or exclude coverage areas, and to reject jamming signals and reduce other electrical interference. The problem of array design, however, is particularly difficult when the antenna elements are closely coupled and are of different lengths. Antenna arrays are usually designed by trial and error because the requirement to include the effects of mutual coupling between the array elements complicates design synthesis. This report describes an approach toward design synthesis that combines a procedure for optimizing feed excitation voltages with the method of moments and which therefore properly includes mutual coupling.

The optimization of excitation voltages to obtain a desired radiation pattern is critical to the design of many antenna systems, especially broadcast antenna systems. Increasingly intense competition for the limited frequency allocation in the AM broadcast band (550-1600 kHz) has led to the use of directional array antennas to reduce interference. In this band, all available channels have been assigned. For a new station to receive a frequency assignment, a guarantee must be made that there will be no interference with existing stations. Thus, new stations must be located a significant distance away from any existing station on that frequency. Often additional protection must be provided by using directional array antennas. In the United States more than 25% of the AM broadcast antennas are directional arrays. This allows considerably more use of each channel and is perhaps the first attempt at practical "frequency reuse" [1].

Typically, the directional antenna design is based on a coverage and protection specification. A typical coverage specification would comprise the following conditions:

- (a) Maximize radiation over the sector covering the city of license.
- (b) Maximize radiation at headings where distant locations are to be protected.

The level of radiation protection (null depth) varies with frequency, distance, and the power level of the local transmitter and the transmitter to be protected. The required width of the nulls depends upon propagation conditions and the geometry of the region to be protected. For daytime coverage, only ground wave propagation must be considered, that is, only nulls at an elevation of 0 degrees. For the night time, because of the existence of sky wave propagation, elevation-pattern nulls must be considered. The directional array antenna design must necessarily combine both quarter wavelength and taller antenna towers to achieve elevation-pattern nulls.

Standard practice in the design of AM broadcast directional array antennas has been to calculate the array excitation voltages, mutual impedances, and patterns on the basis of an assumed sinusoidal current distribution on each antenna. The pattern of the array is then the superposition of the fields from each antenna individually [2]. The problem with this approach is that the currents on the individual antennas are not necessarily sinusoidal. Consequently, the actual mutual impedances are not as predicted. The actual field for each antenna (with the other antennas short-circuited) must account

for scattering from the other antennas. This scattering exists, because a current distribution can exist on an antenna tower even though the input is short-circuited. The sinusoidal current distribution does not consider such interactions. In general, the taller the AM broadcast antenna towers, the more pronounced will be the difference between the actual array patterns and the array patterns calculated on the basis of sinusoidal currents. In practice, the sinusoidal current assumption has in many cases led to incorrect designs [3].

The Mini-Numerical Electromagnetics Code (MININEC) [4] is a computer program which calculates accurate antenna characteristics, including actual current distributions, self- and mutual impedances, and radiation patterns. The procedure described in the following pages for the optimization of the feed excitation voltages of an antenna system to yield a desired radiation pattern makes use of MININEC and is implemented on a microcomputer.

MININEC

MININEC is a frequency-domain method-of-moments computer code for the analysis of thin-wire antennas [4]. MININEC is written in the BASIC language for use on microcomputers with as little as 64K of memory. The MININEC program is based on the numerical solution of an integral representation of the electric fields. A modified Galerkin procedure is used to solve the integral equation [5]. This formulation results in a uniquely compact code (that is, one requiring little core computer storage) suitable for implementation on a microcomputer. The MININEC code solves for the impedance and currents on arbitrarily oriented wires, including configurations with multiple wire junctions. Options include lumped parameter impedance loading of the wires and the calculation of radiation patterns in free space and over flat, perfectly conducting ground planes.

Several options have recently been added to MININEC to enhance its utility for the design and analysis of wire antennas. An approximation based on the Fresnel plane-wave reflection coefficients was included to evaluate the effect of imperfect grounds (impedance not equal to zero) on the radiation pattern. In MININEC the current is computed over a perfect ground, and the radiated field can then be computed over imperfect grounds. Multiple media (that is, changes in surface impedance) can be specified. The choice of medium in the calculation is dependent on the reflection point. A radial ground screen approximation (modified reflection coefficient) can also be used. Diffraction from the edge of the screen is not included.

The analysis of an antenna problem generated by a thin-wire method-of-moments code is at best an approximation. Nonetheless, once the inherent limitations of this approach are taken into account, highly accurate answers can be obtained by carefully modeling the antenna configuration.

POWER GAIN

The radiation pattern specifications of an antenna system are given in terms of power gain or directive gain. Both specifications are given in terms of decibels (dB). The gain for the direction θ and ϕ is

$$G = 4\pi P(\theta, \phi) / P_{in} \quad (1)$$

where $P(\theta, \phi)$ is the power radiated per unit solid angle in the specified direction. P_{in} is the total power to the antenna from the transmitter.

$$P_{in} = \frac{1}{2} R_e (V \cdot I^*) \quad (2)$$

where V is the applied source voltage, and I^* is the conjugate of the resulting feed excitation current. In addition

$$P(\theta, \phi) = \frac{R^2}{2\eta} \bar{E} \cdot \bar{E}^* \quad (3)$$

where R is the radius of the observation sphere; \bar{E} is the far electric field; and η is the intrinsic impedance of free space.

The only difference between directive gain and power gain is that in the latter P_{in} is replaced by P_{rad} , where

$$P_{rad} = P_{in} - P_{loss} \quad (4)$$

and P_{loss} is power loss in the antenna system. The gain in dB is expressed as

$$G_{dB} = 10 \log (G) \quad (5)$$

For the purpose of feed excitation optimization, the multiple feeds of the antenna system are considered to be an N -port network, where N is the number of feeds. The relationship between the port voltages and currents can be expressed as

$$I_i = \sum_{j=1}^N Y_{ij} V_j \quad (6)$$

where Y_{ij} are the traditional short-circuit admittance parameters. The Y -parameters are defined by [6]

$$Y_{ij} = \frac{I_i}{V_j} \left| \begin{array}{l} \text{All } V_k = 0 \\ \text{Except } k = j \end{array} \right. \quad (7)$$

The total field pattern of the antenna system is equal to the sum of the fields due to the excitation of the individual feeds. Changing the magnitude and phase of the individual excitation voltages is the multiplication of the fields from each individual feed (with the other feeds short-circuited) by the appropriate amplitude and phase prior to the summing for the total pattern. Thus, if $e(\theta, \phi)$ is the far electric field due to the implied voltages of $V = 1$, the total far electric field is

$$\bar{E}(\theta, \phi) = \sum_{i=1}^N v_i e_i(\theta, \phi) \quad (8)$$

where v_i would be the actual feed excitation voltages impressed on the antenna system. Substituting equations (2), (3), (7), and (8) into equation (1)

$$G(\theta, \phi) = \frac{4 R^2 \left[\sum_{i=1}^N v_i e_i(\theta, \phi) \right] \cdot \left[\sum_{i=1}^N v_i e_i(\theta, \phi) \right]^*}{\eta \sum_{i=1}^N R_e \left[v_i \sum_{j=1}^N Y_{ij} v_j^* \right]} \quad (9)$$

Thus the power gain can be calculated for any set of feed excitation voltages $e_i(\theta, \phi)$ and the short-circuit admittance parameters, Y_{ij} , are known. These values can be calculated using MININEC.

OPTIMIZATION ALGORITHM

For the purpose of optimizing the excitation voltages of an array antenna to provide for a desired radiation pattern, the following error function is defined:

$$\text{ERROR} = \sum_{j=1}^{NP} w_j(\theta, \phi) [R_j(\theta, \phi) - G_j(\theta, \phi)]^2 \quad (10)$$

where NP is the number of pattern points used to define the required radiation pattern. $R_j(\theta, \phi)$ is the required radiation pattern in dB at the point j. $G_j(\theta, \phi)$ is the achieved antenna pattern, which is given by equation (9) in dB. The error is defined over the entire synthesis range. The least squares error is used to ensure that, when the error is a minimum, each component of the sum is a minimum. Therefore, the best approximation to $R_j(\theta, \phi)$ by $G_j(\theta, \phi)$ is achieved in the mean, and $w_j(\theta, \phi)$ is a weighting function that allows the synthesis precision to be changed over the synthesis range.

The objective of optimization is to minimize the error function of equation (10). A BASIC program, called OPTIMAL, was written for the IBM PC to perform this optimization. The Appendix contains the key variables and a program listing of OPTIMAL.

OPTIMAL uses Newton's method to perform the optimization [7] [8]. Newton's method is the most widely used of all iteration formulas. The method uses extrapolation based on a line that is tangent to the curve. For the case of a single variable, the method can be developed from a Taylor's expansion of the form

$$X_{n+1} = X_n - kF(X_n)/F'(X_n) \quad (11)$$

where X_{n+1} is equivalent to the point where the curve tangent at X_n passes through the X axis. The process is repeated using

$$X_n = X_{n+1} \quad (12)$$

as a new base point. When the value of $F(X_{n+1})$ is sufficiently small the process is terminated. The progress made with each iteration is governed by k , the gradient constant. If k is chosen to be too small, the method will converge very slowly. If k is too large, the process may diverge. In OPTIMAL a best k is chosen using a type of binary search method [7]. First, the function is computed for k equal to a minimum value and k equal to a maximum value. In the case of OPTIMAL the initial maximum value of k was set to .03125, and the initial minimum value of k was set to 0. A new function is calculated using a value of k which is halfway between the two k values. If the new function is greater than the function with the maximum k value and the function with the minimum k value, the minimum k is changed to the mean or halfway k . If the new function is less than the function with the maximum k value and the maximum k function is greater than the function with the minimum k value, the maximum k value is changed to the mean or halfway k value. The process is repeated until the interval between the minimum and maximum k values reaches some minimum value. In the case of OPTIMAL this minimum value is .0000001.

The error function of equation (10) is a function of $2N$ parameters. There is both a real and an imaginary voltage for each of the feed points. Using V to denote each of these $2N$ parameters, each successive iteration of all $2N$ parameters is given by

$$V_{n+1} = V_n - \frac{\text{Error}(V_n)}{\frac{d[\text{Error}(V_n)]}{dV_n}} \quad (13)$$

Analytical expressions are used for each of the required derivatives in OPTIMAL.

SAMPLE PROBLEM

MININEC was used to model an array of three antenna elements. The short-circuit admittance parameters were thus calculated for feeds on each of the three antenna elements. For each calculation, a single feed was impressed with a voltage of $1+j0$ with the other two feeds short-circuited. The far electric field was then calculated for each of the three feeds. Four pattern points for each of the antenna feeds were computed for this example. Additional or fewer pattern points may be chosen depending on the application. As described above, these are the required inputs to the program OPTIMAL. OPTIMAL will read these data from an input file and write the results to an output file. The following is the input file that was used for this sample problem. Comments have been added in italics to identify the data.

```

3                                     :Number of feed points
.120627E-1,-.177122E-1              :Short circuit parameters
.241459E-2,-.164955E-1
.639456E-3,-.14609E-2
.261697E-2,-.256524E-1
.241457E-2,.164956E-1
.120627E-1,-.177122E-1

4                                     :Number of pattern points
0                                     :Radius of observation sphere
2.158E-1,-.380                      :Far electric fields
-.893E-1,1.239
-.557,1.428
-.981,1.446
6.233E-1,3.055E-1
-.246,4.334E-1
1.491E-1,3.752E-1
6.233E-1,3.055E-1
-.981,1.446
3.952E-1,4.996E-1
3.758E-1,5.213E-3
2.158E-1,-.380

10,1                                  :Required gains and weights
-11,1
-40,1
-35,1

250                                   :Maximum number of iterations
4                                     :Minimum least squares error
Y                                     :Option for listing errors
10                                   :Number of iterations before listing

```

The output file which results from the above input file is given below. Editorial remarks that have been added are in italics.

```

*****
      OPTIMAL
      IBM PC VERSION
      10-24-1985
*****

```

The maximum number of feed points is 10. This number can be increased by changing the arrays of lines 120 - 130 of OPTIMAL to the appropriate value. In addition the flag of line 650 would have to be changed.

```

NUMBER OF FEED POINTS? 3
ANTENNA ADMITTANCE VALUES
REAL AND IMAG ADMITTANCE Y( 1 , 1 )? .0120627 ,-.0177122
REAL AND IMAG ADMITTANCE Y( 1 , 2 )? 2.41459E-03 , .0164955
REAL AND IMAG ADMITTANCE Y( 1 , 3 )? 6.39456E-04 ,-.0014609
REAL AND IMAG ADMITTANCE Y( 2 , 2 )? 2.61697E-03 , .0256524
REAL AND IMAG ADMITTANCE Y( 2 , 3 )? 2.41457E-03 , .0164956
REAL AND IMAG ADMITTANCE Y( 3 , 3 )? .0120627 ,-.0177122

```

The maximum number of pattern points is 20. This number can be increased by changing the arrays of lines 120-130 of OPTIMAL to the appropriate value. In addition, the flag of line 920 would have to be changed.

NUMBER OF PATTERN POINTS? 4

When the observation sphere radius is set equal to zero, the effect of range is ignored. This is similar to the convention used in MININEC in the calculation of pattern gains.

OBSERVATION SPHERE RADIUS (M)? 0

PATTERNS (REAL,IMAG) OF INDIVIDUAL FEEDS (V/M)

IT IS ASSUMED THAT FEED VOLTAGES WERE (1,0).

PATTERN POINT	1	OF FEED POINT 1 ?	.2158	,-.38
PATTERN POINT	2	OF FEED POINT 1 ?	-.0893	, 1.239
PATTERN POINT	3	OF FEED POINT 1 ?	-.557	, 1.428
PATTERN POINT	4	OF FEED POINT 1 ?	-.981	, 1.446
PATTERN POINT	1	OF FEED POINT 2 ?	.6233	, .3055
PATTERN POINT	2	OF FEED POINT 2 ?	-.246	, .4334
PATTERN POINT	3	OF FEED POINT 2 ?	.1491	, .3752
PATTERN POINT	4	OF FEED POINT 2 ?	.6233	, .3055
PATTERN POINT	1	OF FEED POINT 3 ?	-.981	, 1.446
PATTERN POINT	2	OF FEED POINT 3 ?	.3952	, .4996
PATTERN POINT	3	OF FEED POINT 3 ?	.3758	, .005213
PATTERN POINT	4	OF FEED POINT 3 ?	.2158	,-.38

DESIRED RADIATION PATTERN (DB)

PATTERN AND WEIGHT OF POINT	1 ?	10	, 1
PATTERN AND WEIGHT OF POINT	2 ?	-11	, 1
PATTERN AND WEIGHT OF POINT	3 ?	-40	, 1
PATTERN AND WEIGHT OF POINT	4 ?	-35	, 1

NUMBER OF ITERATIONS? 250

MINIMUM LEAST SQUARES ERROR? 4

LISTING OF ITERATIONS AND ERROR (Y/N)? Y

NUMBER OF ITERATIONS BEFORE LISTING? 10

Initially, OPTIMAL is slow in converging. However, once it starts to converge, it converges quickly. On the IBM PC the time of the following calculation is greater than 15 minutes. The time of any calculation is dependent on the number of feed points and pattern points. The number of iterations required for convergence is dependent on the word length of the calculations. For shorter word lengths, it will take more iterations for convergence.

ITERATION	ERROR
10	1233.411
20	1222.322
30	1219.684
40	1197.274
50	1193.403
60	1188.957
70	1184.84
80	1182.973
90	1181.16

100	1178.017
110	1172.157
120	1165.476
130	1159.169
140	1157.159
150	1154.309
160	1151.327
170	1149.454
180	1142.514
190	1134.836
200	887.8659
210	382.5797
220	3.573898

***** OUTPUT *****

The minimum least squares error is achieved in 220 trials.

NUMBER OF ITERATIONS = 220

LEAST SQUARES ERROR = 3.573898

***** SYNTHESIZED FEED POINT VOLTAGES *****

FEED NO.	REAL (V)	IMAG (V)	MAGNITUDE (V)	PHASE (DEG)
1	-7.468426E-02	.2037476	.2170042	110.1306
2	1.093298	1.2362	1.6503	48.51037
3	1.480106	-1.013934	1.794095	-34.41275

The above-synthesized feed point voltages were used in MININEC and confirmed the radiation pattern predicted below.

***** RADIATION PATTERN (DB) *****

PATTERN NO.	REQUIRED (DB)	SYNTHESIZED (DB)
1	10	11.08301
2	-11	-9.60534
3	-40	-40.50869
4	-35	-34.55599

INPUT POWER (WATTS) = 2.435024E-02

It is possible with OPTIMAL to continue the synthesis process.

SUMMARY AND RECOMMENDATIONS

A microcomputer can use the program MININEC and OPTIMAL to optimize the antenna system feeds for a desired radiation pattern. This capability demonstrates the growing importance of the microcomputer in antenna modeling.

Several areas for further study can be suggested. The Newton method is slow and does not guarantee that local minimums are being avoided. More modern methods of optimization should be considered for this problem. A conjugate gradient method such as described by Fletcher and Powell [9] or the new least squares algorithm of Dennis et al. [10] are techniques that should be considered. In this study only optimization of feed excitation voltages was considered. It is also possible to optimize antenna locations

and heights. Such an optimization capability would approach a total design capability for the broadcast directional array antenna system and may have useful applications to other directional antenna array systems.

REFERENCES

1. Stafford, E. L., A Guide to Radio and TV Broadcast Engineering Practice, 1971, 1st Edition, Tab Books, Blue Ridge Summit, Pa, 19214.
2. Barlett, G. W. (Editor), National Association of Broadcasters Engineering Handbook. Sixth Edition, Published by National Association of Broadcasters, 1771 N. Street, N. W. Washington, DC, 20036.
3. Sather, O. S., "Construction of the WOR Transmitter and Directional Antenna", IEEE Transactions on Broadcasting, Vol. BC-15, No. 3, September 1969.
4. Li, S. T., et al., Microcomputer Tools for Communication Engineering, Artech House, Inc., Dedham Massachusetts, 1983.
5. Wilton, D. R., "Wire Problems", Lecture Notes for Short Course on Computation Methods in Electromagnetics, 1981.
6. Aaby, P. R., Applied Circuit Theory: Matrix and Computer Methods, John Wiley and Sons, New York, 1980.
7. Shoup, T. E., Applied Numerical Methods for the Microcomputer, Prentice-Hall, Englewood Cliffs, New Jersey, 1984.
8. J. Kowalik and M. R. Osborne, Methods for Unconstrained Optimization Problems, American Elsevier Publishing Company, Inc., New York, 1968.
9. R. Fletcher and M. D. Powell, "A Rapidly Convergent Descent Method for Minimization," The Computer Journal, Vol. 6, 1963, pp. 163-168
10. J. E. Dennis, Jr., D. M. Gray, R. E. Welsch, "An Adaptive Nonlinear Least-squares Algorithm; Technical Report TR-20, Mathematics Research Center, University of Wisconsin, Oct. 1979.

APPENDIX

KEY VARIABLES AND PROGRAM LISTING OF OPTIMAL

Included in this Appendix are the definitions of the key variables and the program listing of the BASIC program OPTIMAL.

LISTING OF KEY VARIABLES

- A(I) - Achieved gain (dB) at pattern point I
- AI(I,J) - Imaginary value of far electric fields (V/m) from feed point I with value of (1,0) at pattern point J
- AR (I,J) - Real value of far electric field (V/m) from feed point I with value of (1,0) at pattern point J
- CG - Maximum constant for gradient
- CS - Flag for continuation of iterations
- EI(I,J) - Imaginary value of far electric field (V/m) from feed point I for actual voltages at pattern point J
- ER(I,J) - Real value of far electric field (V/m) from feed point I for actual voltages at pattern point J
- FI(I) - Imaginary value of voltage at feed point I before iterations
- FR(I) - Real value of voltage at feed point I before iteration
- F\$ - Names of files
- GC - $(2 * \pi) / (\text{intrinsic free space impedance})$
- G(I,1) - Least squares derivative for real value of feed point I
- G(I,2) - Least squares derivative for imaginary value of feed point I
- IT - Number of iterations
- KO - Minimum interval for binary search
- I\$ - Flag for using input file
- L - Changes natural log to log base 10
- LR - Log to base 10 of sphere radius squared (dB)
- MG - Magnitude of feed point voltage (V)
- NF - Number of feed points

NI - Number of iterations
 NL - Number of iterations between listing errors
 NP - Number of pattern points
 OS - Flag for using output file
 P - Phase of feed excitation voltage (deg)
 POWER - Input power (W)
 PO - Changes degrees to radians
 RAD - Observation sphere radius (m)
 R(I) - Required gain (dB) at pattern point I
 SO - Least squares error
 T(I) - Difference in required and achieved gain at pattern point I (dB)
 TT(I) - Magnitude of far electric field (V/m) due to actual feed excitation voltages at pattern point I
 VI(I) - Imaginary value of synthesized voltage at feed point I
 VR(I) - Real value of synthesized voltage at feed point I
 W(I) - Weighted value of least squares error at pattern point I
 YI(I,I) - Imaginary part of short-circuit admittance (mhos)
 YR(I,I) - Real part of short-circuit admittance (mhos)

PROGRAM LISTING

The following BASIC program, OPTIMAL, was written to run on an IBM-PC. However, OPTIMAL has been run on several IBM-compatible computers. With only minor changes it should be possible to run OPTIMAL on most computers.

```

100 REM ----- NUMBER OF FEED POINTS = 10
110 REM ----- NUMBER OF PATTERN POINTS = 20
120 DIM A(20),AR(10,20),AI(10,20),ER(20),EI(20),FR(10),FI(10),G(10,2)
130 DIM R(20),W(20),T(20),TT(20),VR(10),VI(10),YR(10,10),YI(10,10)
140 REM ***** DATA INITIALIZATION *****
150 REM ----- NO CONTINUATION OF ITERATIONS
160 C$="N"
170 REM ----- CHANGES RADIANS TO DEGREES
180 PO=ATN(1)/45
190 REM ----- CHANGES NATURAL LOG TO LOG BASE 10
200 L=10/LOG(10)
210 REM ----- MAXIMUM CONSTANT FOR GRADIENT
220 CG=.03125
230 REM ----- MINIMUM INTERVAL FOR BINARY SEARCH
240 K0=.0000001
250 REM ----- (2 * PI)/(INTRINSIC FREE SPACE IMPEDANCE)
260 GC=.016678
270 REM ***** INPUT/OUTPUT FILES *****
280 PRINT "INPUT FILE (Y/N) ";
290 INPUT I$
300 IF I$="N" THEN 350
310 IF I$<>"Y" THEN 280
320 PRINT "NAME OF FILE ";
330 INPUT F$
340 OPEN F$ FOR INPUT AS #1
350 PRINT "OUTPUT FILE (Y/N)";
360 INPUT O$
370 IF O$="N" THEN 420
380 IF O$<>"Y" THEN 350
390 PRINT "NAME OF FILE ";
400 INPUT F$
410 OPEN F$ FOR OUTPUT AS #2
420 REM ***** TITLE *****
430 W$="*****"
440 X$=" OPTIMAL"
450 Y$=" IBM PC VERSION"
460 Z$=" "+DATES
470 PRINT W$
480 PRINT X$
490 PRINT Y$
500 PRINT Z$
510 PRINT W$
520 PRINT
530 IF O$="Y" THEN PRINT #2,W$
540 IF O$="Y" THEN PRINT #2,X$
550 IF O$="Y" THEN PRINT #2,Y$
560 IF O$="Y" THEN PRINT #2,Z$

```

```

570 IF O$="Y" THEN PRINT #2,W$
580 IF O$="Y" THEN PRINT #2,
590 REM ***** DATA INPUT *****
600 Q$="? "
610 REM ----- ANTENNA SHORT-CIRCUIT ADMITTANCE VALUES
620 W$=" NUMBER OF FEED POINTS"
630 PRINT W$;
640 IF I$="N" THEN INPUT NF ELSE INPUT #1,NF
650 IF NF<11 THEN 690
660 PRINT
670 PRINT "NUMBER OF FEED POINTS EXCEEDS DIMENSION...."
680 IF I$="Y" THEN STOP ELSE 620
690 IF O$="Y" THEN PRINT #2,W$+Q$;NF
700 W$=" ANTENNA ADMITTANCE VALUES"
710 PRINT W$
720 IF O$="Y" THEN PRINT #2,W$
730 FOR I=1 TO NF
740 FOR J=I TO NF
750 W$="REAL AND IMAG ADMITTANCE Y("
760 PRINT W$;I;",";J;")";
770 IF I$="N" THEN INPUT YR(I,J),YI(I,J)
780 IF I$="Y" THEN INPUT #1,YR(I,J),YI(I,J)
790 IF O$="Y" THEN PRINT #2,W$;I;",";J;");Q$;YR(I,J);",";YI(I,J)
800 REM ----- THE ADMITTANCE VALUES ARE SYMMETRICAL
810 IF I=J THEN 840
820 YR(J,I)=YR(I,J)
830 YI(J,I)=YI(I,J)
840 NEXT J
850 NEXT I
860 PRINT
870 IF O$="Y" THEN PRINT #2,
880 REM ----- FAR ELECTRIC FIELD FOR UNIT VOLTAGES
890 W$=" NUMBER OF PATTERN POINTS"
900 PRINT W$;
910 IF I$="N" THEN INPUT NP ELSE INPUT #1,NP
920 IF NP<21 THEN 960
930 PRINT
940 PRINT "NUMBER OF PATTERN POINTS EXCEEDS DIMENSIONS..."
950 IF I$="Y" THEN STOP ELSE 910
960 IF O$="Y" THEN PRINT #2,W$+Q$;NP
970 REM ----- OBSERVATION SPHERE RADIUS
980 W$="OBSERVATION SPHERE RADIUS (M)"
990 PRINT W$;
1000 IF I$="N" THEN INPUT RAD ELSE INPUT #1,RAD
1010 IF O$="Y" THEN PRINT #2,W$+Q$;RAD
1020 REM ----- LOG VALUE TO BE USED IN ACHIEVED GAIN CALCULATION
1030 LR=0
1040 IF RAD>0 THEN LR=2*L*LOG(RAD)
1050 W$=" PATTERNS (REAL,IMAG) OF INDIVIDUAL FEEDS (V/M)"
1060 PRINT W$
1070 IF O$="Y" THEN PRINT #2,W$
1080 W$=" IT IS ASSUMED THAT FEED VOLTAGES WERE (1,0). "
1090 PRINT W$
1100 IF O$="Y" THEN PRINT #2,W$

```

```

1110 FOR I=1 TO NF
1120 FOR J=1 TO NP
1130 W$="PATTERN POINT "
1140 X$=" OF FEED POINT "
1150 PRINT W$;J;X$;I;
1160 IF I$="N" THEN INPUT AR(I,J),AI(I,J)
1170 IF I$="Y" THEN INPUT #1,AR(I,J),AI(I,J)
1180 IF O$="Y" THEN PRINT #2,W$;J;X$;I;Q$;AR(I,J);", ";AI(I,J)
1190 NEXT J
1200 NEXT I
1210 REM ----- DESIRED RADIATION PATTERN (POWER GAIN)
1220 PRINT
1230 IF O$="Y" THEN PRINT #2,
1240 W$=" DESIRED RADIATION PATTERN (DB)"
1250 PRINT W$
1260 IF O$="Y" THEN PRINT #2,W$
1270 FOR I=1 TO NP
1280 W$="PATTERN AND WEIGHT OF POINT "
1290 PRINT W$;I;
1300 IF I$="N" THEN INPUT R(I),W(I) ELSE INPUT #1,R(I),W(I)
1310 IF O$="Y" THEN PRINT #2,W$;I;Q$;R(I);", ";W(I)
1320 NEXT I
1330 PRINT
1340 IF O$="Y" THEN PRINT #2,
1350 REM ----- NUMBER OF ITERATIONS
1360 W$=" NUMBER OF ITERATIONS"
1370 PRINT W$;
1380 IF I$="N" THEN INPUT NI ELSE INPUT #1,NI
1390 IF O$="Y" THEN PRINT #2,W$+Q$;NI
1400 IF C$="Y" THEN NI=NI+IT
1410 REM ----- MINIMUM LEAST SQUARES ERROR
1420 W$=" MINIMUM LEAST SQUARES ERROR"
1430 PRINT W$;
1440 IF I$="N" THEN INPUT ER ELSE INPUT #1,ER
1450 IF O$="Y" THEN PRINT #2,W$+Q$;ER
1460 REM ----- OPTION OF LISTING ITERATIONS AND ERROR
1470 W$="LISTING OF ITERATIONS AND ERROR (Y/N)"
1480 PRINT W$;
1490 IF I$="N" THEN INPUT L$ ELSE INPUT #1,L$
1500 IF O$="Y" THEN PRINT #2,W$+Q$;L$
1510 IF L$="N" THEN 1570
1520 W$=" NUMBER OF ITERATIONS BEFORE LISTING"
1530 PRINT W$;
1540 IF I$="N" THEN INPUT NL ELSE INPUT #1,NL
1550 IF O$="Y" THEN PRINT #2,W$+Q$;NL
1560 IF NL=0 THEN L$="N"
1570 PRINT
1580 IF O$="Y" THEN PRINT #2,
1590 IF L$="Y" THEN PRINT " ", "ITERATION", "ERROR"
1600 IF L$="Y" AND O$="Y" THEN PRINT #2, " ", "ITERATION", "ERROR"
1610 IF C$="Y" THEN 1780
1620 REM ***** BEGIN ITERATIONS *****
1630 IT=0
1640 REM ----- IF FEED VOLTAGES ARE NOT (1,0) FOR PATTERN, CHANGE

```

```

1650 FOR I=1 TO NF
1660 VR(I)=1
1670 VI(I)=0
1680 NEXT I
1690 REM ----- COMPUTE FIRST ACHIEVED GAIN AND ERROR
1700 GOSUB 3060
1710 REM ----- FIRST ERROR
1720 S0=S
1730 REM ***** OUTPUT CHECK *****
1740 REM ----- IF ITERATIONS EQUAL MAXIMUM,
1750 IF IT=NI THEN 2380
1760 REM ----- IF COMPUTED ERROR IS LESS THAN MINIMUM ERROR,
1770 IF S0<=ER THEN 2380
1780 REM ***** ERROR DERIVATIVE *****
1790 REM ----- CONSIDER ALL FEED POINTS
1800 FOR K=1 TO NF
1810 REM ----- CONSIDER BOTH REAL AND IMAGINARY VOLTAGES
1820 FOR II=1 TO 2
1830 D1=0
1840 REM ----- INPUT POWER DERIVATIVE
1850 FOR J=1 TO NF
1860 REM ----- FUNCTION OF REAL VOLTAGE
1870 IF II=1 THEN D1=D1+VR(J)*YR(K,J)
1880 REM ----- FUNCTION OF IMAGINARY VOLTAGE
1890 IF II=2 THEN D1=D1+VI(J)*YR(K,J)
1900 NEXT J
1910 D=0
1920 FOR I=1 TO NP
1930 REM ----- ELECTRIC FIELD DERIVATIVES
1940 IF II=1 THEN D2=2*(ER(I)*AR(K,I)+EI(I)*AI(K,I))
1950 IF II=2 THEN D2=2*(EI(I)*AR(K,I)-ER(I)*AI(K,I))
1960 D=D+W(I)*T(I)*(D1/POWER-D2/TT(I))
1970 NEXT I
1980 REM ----- LEAST SQUARES DERIVATIVE
1990 G(K,II)=2*L*D
2000 NEXT II
2010 NEXT K
2020 REM ***** NEWTON METHOD *****
2030 REM ----- SAVE PRESENT VOLTAGES
2040 FOR I=1 TO NF
2050 FR(I)=VR(I)
2060 FI(I)=VI(I)
2070 NEXT I
2080 REM ----- ITERATIVE METHOD OF FINDING GAIN CONSTANTS
2090 C0=0
2100 C=CG
2110 REM ----- ERROR CALCULATION FOR FIRST GRADIENT CONSTANT
2120 GOSUB 3010
2130 C2=CG
2140 S2=S
2150 REM ----- BINARY SEARCH FOR BEST GRADIENT CONSTANT
2160 C=(C2+C0)/2
2170 REM ----- ERROR CALCULATION FOR SUCCESSIVE GRADIENT CONSTANTS
2180 GOSUB 3010

```

```

2190 S1=S
2200 C1=C
2210 IF S2<S1 AND S0>S2 THEN 2250
2220 C2=C1
2230 S2=S1
2240 GOTO 2270
2250 C0=C1
2260 S0=S1
2270 IF C2-C0>K0 THEN 2160
2280 REM ----- ERROR CALCULATION AFTER BINARY SEARCH FOR GRADIENT CONSTANT
2290 S0=S1
2300 REM ***** UPDATE ITERATION *****
2310 IT=IT+1
2320 REM ----- ITERATION AND ERROR LISTING OPTION
2330 IF L$="N" THEN 1750
2340 TI=NL*INT(IT/NL)
2350 IF IT=TI THEN PRINT " ",IT,S0
2360 IF IT=TI AND O$="Y" THEN PRINT #2," ",IT,S0
2370 GOTO 1750
2380 REM ***** OUTPUT *****
2390 W$="*****" OUTPUT "*****"
2400 PRINT W$
2410 PRINT
2420 IF O$="Y" THEN PRINT #2,W$
2430 IF O$="Y" THEN PRINT #2,
2440 REM ----- NUMBER OF ITERATIONS
2450 W$="NUMBER OF ITERATIONS ="
2460 PRINT W$;IT
2470 IF O$="Y" THEN PRINT #2,W$;IT
2480 REM ----- LEAST SQUARES ERROR
2490 W$="LEAST SQUARES ERROR ="
2500 PRINT W$;S0
2510 IF O$="Y" THEN PRINT #2,W$;S0
2520 PRINT
2530 IF O$="Y" THEN PRINT #2,
2540 REM ----- SYNTHESIZED FEED POINT VOTAGES
2550 W$="***** SYNTHESIZED FEED POINT VOLTAGES *****"
2560 PRINT " ",W$
2570 IF O$="Y" THEN PRINT #2," ",W$
2580 U$="FEED NO."
2590 W$="REAL (V)"
2600 X$="IMAG (V)"
2610 Y$="MAGNITUDE (V)"
2620 Z$="PHASE (DEG)"
2630 PRINT U$,W$,X$,Y$,Z$
2640 IF O$="Y" THEN PRINT #2,U$,W$,X$,Y$,Z$
2650 FOR I=1 TO NF
2660 REM ----- MAGNITUDE AND PHASE CALCULATIONS
2670 MG=SQR(VR(I)*VR(I)+VI(I)*VI(I))
2680 IF VR(I)<>0 THEN 2710
2690 P=0
2700 GOTO 2730
2710 P=ATN(VI(I)/VR(I))/PO

```

```

2720 IF VR(I)<0 THEN P=P+SGN(VI(I))*180
2730 PRINT I,VR(I),VI(I),MG,P
2740 IF O$="Y" THEN PRINT #2,I,VR(I),VI(I),MG,P
2750 NEXT I
2760 PRINT
2770 IF O$="Y" THEN PRINT #2,
2780 REM ----- RADIATION PATTERNS
2790 W$="          ***** RADIATION PATTERN (DB) *****"
2800 PRINT W$
2810 IF O$="Y" THEN PRINT #2,W$
2820 W$="PATTERN NO."
2830 X$="REQUIRED (DB)"
2840 Y$="SYNTHESIZED (DB)"
2850 PRINT W$,X$,Y$
2860 IF O$="Y" THEN PRINT #2,W$,X$,Y$
2870 FOR I=1 TO NP
2880 PRINT I,R(I),A(I)
2890 IF O$="Y" THEN PRINT #2,I,R(I),A(I)
2900 NEXT I
2910 REM ----- INPUT POWER
2920 W$=" INPUT POWER (WATTS)="
2930 PRINT W$;POWER
2940 IF O$="Y" THEN PRINT #2,W$;POWER
2950 PRINT
2960 REM ***** CONTINUE ITERATIONS *****
2970 PRINT "CONTINUE ITERATIONS (Y/N)";
2980 INPUT C$
2990 IF C$="Y" THEN 1330
3000 GOTO 3400
3010 REM ***** NEW VOLTGAES *****
3020 FOR I=1 TO NF
3030 VR(I)=FR(I)-C*G(I,1)
3040 VI(I)=FI(I)-C*G(I,2)
3050 NEXT I
3060 REM ***** ACHIEVED GAIN *****
3070 REM ----- ELECTRIC FIELD FOR ACTUAL VOLTAGES
3080 FOR J=1 TO NP
3090 ER(J)=0
3100 EI(J)=0
3110 FOR I=1 TO NF
3120 ER(J)=ER(J)+VR(I)*AR(I,J)-VI(I)*AI(I,J)
3130 EI(J)=EI(J)+VR(I)*AI(I,J)+VI(I)*AR(I,J)
3140 NEXT I
3150 NEXT J
3160 REM ----- INPUT POWER CALCULATION FOR ACTUAL VOLTAGES
3170 POWER=0
3180 FOR I=1 TO NF
3190 P1=0
3200 P2=0
3210 FOR J=1 TO NP
3220 P1=P1+YR(I,J)*VR(J)-YI(I,J)*VI(J)
3230 P2=P2+YI(I,J)*VR(J)+VI(J)*YR(I,J)
3240 NEXT J

```

```
3250 POWER=POWER+VR(I)*P1+VI(I)*P2
3260 NEXT I
3270 POWER=POWER/2
3280 REM ***** LEAST SQUARES ERROR *****
3290 S=0
3300 FOR I=1 TO NP
3310 TT(I)=ER(I)*ER(I)+EI(I)*EI(I)
3320 REM ----- ACHIEVED GAIN (DB) FOR ACTUAL VOLTAGES
3330 A(I)=L*LOG(GC*TT(I)/POWER)+LR
3340 REM ----- DIFFERENCE IN ACHIEVED AND REQUIRED GAIN
3350 T(I)=R(I)-A(I)
3360 REM ----- LEAST SQUARES ERROR WITH WEIGHTING
3370 S=S+W(I)*T(I)*T(I)
3380 NEXT I
3390 RETURN
3400 END
```

INITIAL DISTRIBUTION

U.S. NAVY

NAVAL SPACE & WARFARE SYSTEMS COMMAND
WASHINGTON, DC 20363-5100
CODE 614

NAVAL POSTGRADUATE SCHOOL
MONTEREY, CA 93940
CODE 62AB

U.S. ARMY

USA-CECOM
AMSEL-COM-TN-4
FORT MONMOUTH, NJ 07703

USA-ISESA
ASBN-SET-P
FORT HUACHUCA, AZ 85613

U.S. AIR FORCE

RADC-EEP
HANSCOM AFB
BEDFORD, MA 01731

OTHERS

LAWRENCE LIVERMORE NATIONAL LABORATORY
PO BOX 808
LIVERMORE, CA 94550

UNIVERSITY OF KANSAS
LAWRENCE, KS 66045

SYSTEMS DEVELOPMENT CORPORATION
4025 HANCOCK STREET
SAN DIEGO, CA 92110

DEFENSE TECHNICAL INFORMATION CENTER
ALEXANDRIA, VA 22314

END

FILMED

6-86

DITIC